


Extension

# gm\_graphes

Version 1.0

# Asymptote



# 1 Type GRAPHE

## Structure GRAPHE

```
struct GRAPHE {
    pair[] s_xy;
    string[] s_lab;
    envelope[] s_env;
    pen[] s_penlab;
    pen[] s_penenv;
    filltype[] s_fill;
    real[][] a_k; // Matrice d'adjacence (ou booléenne)
    real[][] a_ang;
    string[][] a_lab;
    real[][] a_labpos;
    align[][] a_labdir;
    pen[][] a_pen;
    pen[][] a_penbelow;
    pen[][] a_penlab;
    arrowbar[][] a_arrow;
    filltype[][] a_labfill;
    string configuration; // ... des sommets
    stylegraphe grstyle;
    ...
}
```

Remarques :

- les paramètres débutant par **s\_** et **a\_** sont respectivement relatifs aux sommets aux arêtes.
- Le nombre de paramètres est important pour permettre un haut degré de personnalisation mais il faut savoir que :
  - un fichier de styles (fourni ou personnel) facilitera la mise en forme des graphes;
  - beaucoup de syntaxes simples de la fonction GRAPHE sont envisagées, pour permettre de créer rapidement des graphes avec un style choisi, et un minimum de renseignements à donner ; certaines syntaxes ne nécessiteront pas de renseigner une matrice d'adjacence.

## 2 Fonction GRAPHE - Syntaxe 1

GRAPHE(**s\_xy**, **a\_k** [, **arg.optionnels**])

Avec cette première syntaxe, sont obligatoires :


- l'argument (**pair[]**) **s\_xy**, qui est l'argument principal qui va définir les sommets du graphe ;
- l'argument (**real[][]**) **a\_k**, qui est la matrice d'adjacence du graphe.

Les arguments optionnels sont :

- (**pair[]**) **s\_lab**, qui permet de nommer les sommets à sa convenance ;
- (**stylegraphe**) **grstyle**, qui est un style de graphe que l'on peut choisir (ou définir) dans le fichier **gm\_graphes\_styles**.

IMPORTANT : **s\_lab** et **a\_k** seront automatiquement tronqués ou complétés pour correspondre à la taille de **s\_xy**.

### 2.1 Un exemple de la syntaxe 1, sans options




Exemple 2 – Syntaxe 1 sans options

```
import gm_graphes;
size(7cm,0);
pair sA=(0,0), sB=(4,3), sC=(5,-2), sD=(-1,3),
     sE=(-2,1), sF=(1,-2), sG=(-2,-1);
pair[] sommets={sA,sB,sC,sD,sE,sF,sG};
real[][] matadj={{1,1,0,0,0,1},
                 {0,0,0,1,0,0},
                 {1,3,0,0,0,1},
                 {0,2,0,0,1,0}};
GRAPHE gr=GRAPHE(sommets, matadj);
draw(gr);
```

(**s\_xy**=) **sommets** étant une liste de 7 paires :

- (**s\_lab**=) **noms**, qui est une liste de 2 labels, va être complétée avec  $s_2, s_3, s_4, s_5, s_6$
- (**arck**=) **matadj**, qui est une matrice 4x5, va être complétée par des zéros pour obtenir une matrice 7x7.

## 2.2 Un autre exemple de la syntaxe 1, avec quelques options



Exemple 3 – Syntaxe 1 avec options

```
import gm_graphes;
import gm_graphes_styles;
size(7cm,0);
pair sA=(0,0), sB=(4,3), sC=(5,-2), sD=(-1,3),
      sE=(-2,1), sF=(1,-3), sG=(-2,-2);
pair [] sommets={sA,sB,sC,sD,sE,sF,sG};
string [] noms={"$A$", "$B$", "$C$", "$D$", "$E$"};
real [] [] matadj={{1,1,0,0,0,1},
                  {0,0,0,1,0,0},
                  {1,3,0,0,0,1},
                  {0,2,0,0,1,0}};
GRAPHE gr=GRAPHE(sommets,
                 noms,
                 matadj,
                 style_boule_1
                 );
// On a adopté le style "style_boule_1"
// mais on modifie le style d'une arête avant traçage.
modif_a(gr,0,1,a_ang=0,a_pen=1.5bp+.5blue,
        a_penbelow=2.5bp+paleblue);
draw(gr,aff_a_lab=true);
```

## 3 Fonction GRAPHE - Syntaxe 2

GRAPHE(**a\_k**, **configuration** [, **arg.optionnels**])

Avec cette deuxième syntaxe, sont obligatoires :


- l'argument (**real[][]**) **a\_k**, qui est la matrice d'adjacence (ou booléenne) du graphe.
- l'argument (**string**) **configuration**, qui permet de spécifier un type de disposition des noeuds (en cercle, en cercle avec centre, ...).

Les arguments optionnels sont :

- (**pair[]**) **s\_lab**, qui permet de nommer les sommets à sa convenance ;
- (**stylegraphe**) **grstyle**, qui est un style de graphe que l'on peut choisir (ou définir) dans le fichier **gm\_graphes\_styles**.

IMPORTANT : **s\_lab** sera automatiquement tronqué ou complété pour correspondre à la taille de **a\_k**; préalablement, si **a\_k** devait être non carré, il sera lui même complété avec une(des) ligne(s) ou colonne(s) de zéro.


### 3.1 Un exemple de la syntaxe 2, sans options



Exemple 4 – Syntaxe 2 sans options

```
import gm_graphes;
size(7cm,0);
real [] [] matadj={{1,1,0,0,0},
                  {0,0,0,1,1},
                  {0,3,0,1,0},
                  {0,2,0,0,4}};
GRAPHE gr=GRAPHE(matadj,configuration="cercle");
draw(gr);
```

### 3.2 Un autre exemple de la syntaxe 2, avec quelques options



Exemple 5 – Syntaxe 2 avec options

```
import gm_graphes;
import gm_graphes_styles;
size(7cm,0);
string[] noms={"$A$", "$B$", "$C$", "$D$", "$E$"};
real[][] matadj={{1,1,0,0,0,1},
                 {0,0,0,1,0,0},
                 {1,3,0,0,0,1},
                 {0,2,0,0,1,0}};
GRAPHE gr=GRAPHE(noms,
                 matadj,
                 configuration="cercle",
                 style_boule_2
                 );
// On a adopté le style "style_boule_2" mais on modifie
// le style d'un sommet et d'une arête avant traçage.
modif_s(gr,2,s_penlab=blue,s_penenv=2bp+red,
        s_fill=FillDraw(yellow));
modif_a(gr,2,5,a_ang=0,a_labpos=.65,
        a_pen=1.5bp+.5blue,a_penbelow=2.5bp+paleblue);
draw(gr,aff_a_lab=true);
```

## 4 Fonction GRAPHE - Syntaxe 3

Il faudra être patient !

## 5 Fonction GRAPHE - Syntaxe 4

Il faudra être patient !

## 6 Fonction GRAPHE - Syntaxe 5

Il faudra être patient !

## 7 Fonction GRAPHE - D'autres syntaxes possibles.

Il faudra être patient !

## 8 D'autres fonctions...

### 8.1 Fonction modif\_s

Une fonction qui, comme son nom tente de l'indiquer, permet de modifier les sommets.

### 8.2 Fonction modif\_a

Une fonction qui, comme son nom tente de l'indiquer, permet de modifier les arêtes.

### 8.3 Fonction draw - Syntaxe 1

`draw(gr [, arg.optionnels])`

Une fonction qui, sans grande originalité, permet de dessiner un graphe préalablement défini.

### 8.4 D'autres fonctions encore...

Il faudra être patient !

## 9 Un aperçu du fichier de style

```
// Des styles prédéfinis de graphes pour l'extension gm_graphes

import gm_graphes;


/////**** Nouveau filltype
/////**** RadialShade avec centre légèrement décalé
filltype EffetBalle(pen penc, pen penr)
{
    return filltype(new void(frame f, path[] g, pen) {
        pair c=(min(g)+max(g))/2+(-4,4);
        radialshade(f,g,penc,c,0,penr,c,abs(max(g)-min(g))/2);
    });
}
/////*****
///// STYLE 0 : stylepardefaut (déplacé dans gm_graphes.asy)
/////*****

/////*****
///// STYLE 1 : style_boule_1
/////*****
stylegraphe style_boule_1
    =stylegraphe(s_env      = ellipse,
                s_penlab   = yellow,
                s_penenv   = currentpen,
                s_fill     = EffetBalle(orange,black),
                a_ang      = -20,
                a_labpos   = 0.55,
                a_labdir   = Relative(E),
                a_pen      = orange,
                a_penbelow = 1bp+black,
                a_penlab   = currentpen,
                a_arrow    = Arrow(10bp),
                a_labfill  = NoFill);

/////*****
///// STYLE 2 : style_boule_2
/////*****
stylegraphe style_boule_2
    =stylegraphe(s_env      = ellipse,
                s_penlab   = blue,
                s_penenv   = currentpen,
                s_fill     = EffetBalle(yellow,paleblue),
                a_ang      = -20,
                a_labpos   = 0.55,
                a_labdir   = Relative(E),
                a_pen      = 1bp+yellow,
                a_penbelow = 1.5bp+black,
                a_penlab   = currentpen,
                a_arrow    = Arrow(SimpleHead,8bp),
                a_labfill  = NoFill);
```


## 10 Des exemples en vrac

... aux pages suivantes!



Exemple 6 – Syntaxe 2 configuration cercle


```
import gm_graphes;
import gm_graphes_styles;
size(7cm,0);
string[] noms={"$A$", "$B$", "$C$", "$D$", "$E$"};
real[] [] matadj={{1,1,0,0,0,1},
                 {0,0,0,1,0,0},
                 {1,3,0,0,0,1},
                 {0,2,0,0,1,0}};
GRAPHE gr=GRAPHE(noms,
                 matadj,
                 configuration="cerclecentre"
                 );
modif_a(gr,2,5,a_ang=0,a_labpos=.7);
draw(gr,aff_a_lab=true);
```



Exemple 7 – Syntaxe 2 configuration cerclecentre

```
import gm_graphes;
import gm_graphes_styles;
size(7cm,0);
string[] noms={"$A$", "$B$", "$C$", "$D$", "$E$"};
real[] [] matadj={{1,1,0,0,0,1},
                 {0,0,0,1,0,0},
                 {1,3,0,0,0,1},
                 {0,2,0,0,1,0}};
GRAPHE gr=GRAPHE(noms,
                 matadj,
                 configuration="cerclecentre",
                 style_boule_2
                 );
modif_s(gr,2,s_penlab=fontsize(18pt)+red,
        s_env=roundbox,
        s_penenv=2bp+.5blue,
        s_fill=RadialShadeDraw(paleblue,white));
modif_a(gr,2,5,a_ang=0,a_labpos=.65,
        a_pen=1bp+.5blue,
        a_penbelow=3bp+paleblue,
        a_lab="yes !",
        a_penlab=fontsize(16pt)+.5green);
draw(gr,aff_a_lab=true);
```

### Exemple 8 – Syntaxe 2 configuration cercle




```
import gm_graphes;
import gm_graphes_styles;
size(7cm,0);
string[] noms={"A","B"};
real[][] matadj={{0,1,0,0,0,1},
                 {0,0,0,1,0,0},
                 {0,0,0,0,1,0},
                 {0,0,0,0,0,1}};

style_boule_1.a_arrow=None;
style_boule_1.a_ang=0;
style_boule_1.a_labpos=0.5;

GRAPHE gr=GRAPHE(noms,
                 matadj,
                 configuration="cercle",
                 style_boule_1
                 );
modif_a(gr,2,4,a_penlab=blue,a_labdir=Relative(W));

draw(gr,aff_a_lab=true);
```

### Exemple 9 – Syntaxe 1 avec diverses options



```
import gm_graphes;
import gm_graphes_styles;
size(7cm,0);
pair sA=(1,4),
     sB=(0,3), sC=(1,3), sD=(2,3), sE=(3,2.5),
     sF=(0,2), sG=(1,2), sH=(2,2),
     sI=(1,1), sJ=(2,0);
pair[] sommets={sA,sB,sC,sD,sE,sF,sG,sH,sI,sJ};
string[] noms={"A","B","C","D","E","F","G","H","I","J"};
real[][] matadj=
    {{ 0, 85,217, 0,173, 0, 0, 0, 0, 0}, //A
     { 85, 0, 0, 0, 0, 80, 0, 0, 0, 0}, //B
     {217, 0, 0, 0, 0, 0,186,103, 0, 0}, //C
     { 0, 0, 0, 0, 0, 0, 0,0,183, 0, 0}, //D
     {173, 0, 0, 0, 0, 0, 0, 0, 0,0,502}, //E
     { 0, 80, 0, 0, 0, 0, 0, 0, 0,250, 0}, //F
     { 0, 0,186, 0, 0, 0, 0, 0, 0, 0, 0}, //G
     { 0, 0,103,183, 0, 0, 0, 0, 0,0,167}, //H
     { 0, 0, 0, 0, 0, 0,250, 0, 0, 0, 84}, //I
     { 0, 0, 0, 0, 0,501, 0, 0,167, 84, 0}}; //J

// On va adopter le style "style_boule_2"
// mais on en modifie préalablement 3 des 10 paramètres.
style_boule_2.a_penlab=fontsize(8pt);
style_boule_2.a_labpos=.5;
style_boule_2.a_penbelow=2bp+blue;

GRAPHE gr=GRAPHE(sommets, noms, matadj, style_boule_2 );

modif_a(gr,0,4,a_ang=10); // Modification de la courbure
modif_a(gr,4,0,a_ang=-40); // de 4 flèches.
modif_a(gr,4,9,a_ang=10);
modif_a(gr,9,4,a_ang=-40);

draw(gr,aff_a_lab=true);
```